

Алгоритмы обработки
натуральных чисел.
Задачи № 22, 25 ЕГЭ 2021

Алгоритм перестановки двух чисел

местами

- С третьей переменной

```
tmp := a;
```

```
a := b;
```

```
b := tmp;
```

- Используя только две переменные

```
a:=a+b;
```

```
  b:=a-b;
```

```
  a:=a-b;
```

```
  a:=a/b;
```

```
  b:=b*a;
```

```
  a:=b/a;
```

22

Ниже на четырёх языках программирования записан алгоритм. Получив на вход число x , этот алгоритм печатает два числа: L и M . Укажите наибольшее число x , при вводе которого алгоритм печатает сначала 4, а потом 5.

```
var x, L, M, Q: integer;
begin
  readln(x);
  Q := 9;
  L := 0;
  while x >= Q do
  begin
    L := L + 1;
    x := x - Q;
  end;
  M := x;
  if M < L then
  begin
    M := L;
    L := x;
  end;
  writeln(L);
  writeln(M);
end.
```

Решение:

По коду из числа x четыре раза вычитается 9. В X остается значение < 9 .

Т.е. $x - 9 * L = M$

$X = 4 + 5 * 9 = 49$ или $X = 5 + 4 * 9 = 41$

Ответ: 49

Программа

```
var x,l,m,q,i:integer;
begin
for i:=1000 downto 1 do begin
  x:=i;
  q:=9;
  l:=0;
  while x>=q do begin
    l:=l+1;
    x:=x-q;
  end;
  m:=x;
  if m<l then begin m:=l; l:=x; end;
  if (l=4) and (m=5) then begin writeln(i); break end;
  end;
end.
```

Целочисленное деление

$n \bmod 10$

$n \operatorname{div} 10$

$n \bmod 2 = 0$

$n \bmod x \neq 0$

Дано трехзначное число, найти сумму его цифр.

```
readln(n);
```

```
s:= n div 100 + n div 10 mod 10 + n mod 10;
```

```
writeln(s);
```

Разложение натурального числа на цифры

Идея	
n	цифра
34562	2
3456	6
345	5
34	4
3	3
0	

```
var n:integer;  
begin  
  readln(n);  
  while n>0 do begin  
    write(n mod 10, ' ');  
    n:=n div 10;  
  end;  
end.
```

Вместо вывода цифр может быть
s:= s + n mod 10;
k:= k+1;
и т.п.

Представление натурального числа в системе счисления по основанию p ($p \leq 10$)

Идея	
$n = n \text{ div } 8$	$n \text{ mod } 8$
3456	0
432	0
54	6
6	6
0	

```
var n:integer;
begin
  readln(n);
  while n>0 do begin
    write(n mod p, ' ');
    n:=n div p;
    end;
end.
```

Вместо вывода цифр может быть
 $s := s + n \text{ mod } p$;
 $k := k + 1$;
и т.п.

1. Ниже записан алгоритм. Сколько существует таких чисел, при вводе которых алгоритм печатает сначала 2, а потом 12?

```
var x, a, b: integer;  
begin  
  readln(x);  
  a:=0; b:=0;  
  while x>0 do begin  
    a:=a + 1;  
    b:=b + (x mod 10);  
    x:=x div 10;  
  end;  
  writeln(a);  
  write(b);  
end.
```

Решение:

а – количество цифр,
b – сумма цифр
X – двузначное число,
сумма цифр = 12
Это числа
39,48,57,66,75,84,93
Всего 7 чисел.

Программа

```
var x, a, b, i, k: integer;
begin
  k:=0;
  for i:=1 to 1000 do begin
    x:=i;
    a:=0; b:=0;
    while x>0 do begin
      a:=a + 1;
      b:=b + (x mod 10);
      x:=x div 10;
      end;
      if (a=2) and (b=12) then k:=k+1;
      end;
    writeln(k);
  end.
```

2. Ниже записана программа. Получив на вход число x , эта программа печатает два числа, L и M . Укажите наибольшее из таких чисел x , при вводе которых алгоритм печатает сначала 3, а потом 7.

```
var x, L, M: integer;
begin
  readln(x);
  L := 0;
  M := 0;
  while x > 0 do
    begin
      L := L + 1;
      if x mod 2 = 0 then
        M := M + (x mod 10) div 2;
      x := x div 10;
    end;
  writeln(L);
  writeln(M);
end.
```

Решение:

L – количество цифр, M – сумма половинок четных цифр.

x - число трехзначное, сумма половинок его четных цифр = 7.

Наибольшее число 986.

Программа

```
var x, L, M, max,i: integer;
begin
max:=0;
for i:=100 to 1000 do begin
  x:=i;
  L := 0;
  M := 0;
  while x > 0 do begin
    L := L + 1;
    if x mod 2 = 0 then
      M := M + (x mod 10) div 2;
    x := x div 10;
  end;
  if (L=3)and(M=7) {and(i>max)} then max:=i;
  end;
  writeln(max);
end.
```

3. Получив на вход натуральное число x , этот алгоритм печатает два числа: a и b . Укажите **наименьшее** натуральное число, при вводе которого алгоритм печатает сначала 4, а потом – 5

```
var x, a, b: longint;  
begin  
  readln(x);  
  a := 0; b := 1;  
  while x > 0 do begin  
    if x mod 2 > 0 then  
      a := a + x mod 9  
    else  
      b := b * (x mod 9);  
    x := x div 9;  
  end;  
  writeln(a);  
  write(b);  
end.
```

Решение: 1) Число в сс с нечетным основанием чётно тогда и только тогда, когда сумма всех его цифр чётна. 2) Четность числа сохраняется в любой сс.

a – сумма цифр в 9 сс, b – произведение цифр в 9 сс. $a=4$, $b=5$

$a=4=1+3=2+2=\cancel{1+1+1+1}=\cancel{1+1+2}$ - ищем наименьшее число. $b=5$ – одна цифра.

Проверим 45. Число нечетное, $a=5$ – неверно!

Проверим 54. Число нечетное, $a=4$, $x=5$, снова число нечетное $a=9$ – неверно!

Проверим число 135. Число нечетное $\rightarrow a=5$, не подходит!

Пусть число 153. Число нечетное $\rightarrow a=3$, число 15 четное $\rightarrow b=5$, число 1 нечетное $\rightarrow a=4$

Ответ: $153_9=129$

Программа

```
var x, a, b,i: longint;  
begin  
for i:=1 to 1000 do begin  
    x:=i;  
    a := 0; b := 1;  
    while x > 0 do begin  
        if x mod 2 > 0 then  
            a := a + x mod 9  
        else  
            b := b * (x mod 9);  
            x := x div 9;  
        end;  
        if (a=4) and(b=5) then begin writeln(i); break; end;  
    end;  
end.  
end.
```

Алгоритм Евклида (НОД)

1-й вариант:

```
while L <> M do
  if L > M then
    L := L - M
  else
    M := M - L;
```

12	33
12	21
12	9
3	9
3	6
3	3

2-й вариант:

```
while (L <> 0) and (M <> 0) do
  if L > M then L := L mod M
  else M := M mod L;
writeln (L + M);
```

12	33
12	9
3	9
3	0

4. Ниже на пяти языках программирования записан алгоритм. Получив на вход число x , этот алгоритм печатает число M . Известно, что $x > 100$. Укажите наименьшее такое (т.е. большее 100) число x , при вводе которого алгоритм печатает 26.

```

var x, L, M: integer;
begin
  readln(x);
  L := x;
  M := 65;
  if L mod 2 = 0 then M := 52;
  while L <> M do
    if L > M then L := L - M
      else M := M - L;
  writeln(M);
end.

```

Решение:

НОД=26

Тогда $M=52$, L должно быть четное и >100

НОД(L,M)=26

Пусть $L=104$

52	104
52	52

Пусть $L=130$

52	130
52	78
52	26
26	26

Ответ: 26

Программа

```
var x, L, M,i: integer;
begin
for i:=100 to 1000 do begin
    x:=i;
    L := x;
    M := 65;
    if L mod 2 = 0 then M := 52;
    while L <> M do
        if L > M then
            L := L - M
            else M := M-L;
        if M=26 then begin writeln(i); break; end;
    end;
end.
```


5. Ниже записан алгоритм. После выполнения алгоритма было напечатано 3 числа. Первые два напечатанных числа – это числа 9 и 81. Какое наибольшее число может быть напечатано третьим?

```
var x, y, z: longint;  
    r, a, b: longint;  
begin  
    readln(x, y);  
    if y > x then begin  
        z:= x; x:= y; y:= z;  
    end;  
    a:= x; b:= y;  
    while b > 0 do begin  
        r:= a mod b;  
        a:= b;  
        b:= r;  
    end;  
    writeln(a);  
    writeln(x);  
    write(y);  
end.
```

Решение:
НОД(x,y)=9
x>y по коду!

x=81, пусть y=72

a	b	r
81	72	9
72	9	0
9	0	

Программа

```
var x, y, z: integer; r, a, b, max, i, j: integer;
begin
  max:=0;
  for i:=1 to 200 do
    for j:=1 to 200 do begin
      x:=i; y:=j;
      if y > x then begin
        z:= x; x:= y; y:= z; end;
      a:= x; b:= y;
      while b > 0 do begin
        r:= a mod b;
        a:= b;
        b:= r;
      end;
      if (a=9) and (x=81) {and (y>max)} then max:=y
        end;
    writeln(max);
  end.
```

Делители натурального числа

1 способ:

```
readln(n);
```

```
k:=0;
```

```
for i:=1 to n do
```

```
    if n mod i =0 then k:=k+1;
```

или

```
readln(n);
```

```
k:=2;
```

```
for i:=2 to n div 2 do
```

```
    if n mod i =0 then k:=k+1;
```

Делители натурального числа

2 способ:

readln(n);

k:=0;

for i:=1 to trunc(sqrt(n)) do

 if n mod i =0 then

 if i*i=n then k:=k+1 {делитель=i}

 else k:=k+2; {два делителя: i и n div i}

Напишите программу, которая ищет среди целых чисел, принадлежащих числовому отрезку [174457; 174505], числа, имеющие ровно два различных натуральных делителя, не считая единицы и самого числа. Для каждого найденного числа запишите эти два делителя в таблицу на экране с новой строки в порядке возрастания произведения этих двух делителей. Делители в строке таблицы также должны следовать в порядке возрастания.

```
var i,k,j:integer;
a:array[1..2] of integer;
begin
for i:=174457 to 174505 do begin
k:=0;
for j:=2 to i div 2 do
    if i mod j=0 then if k<2 then begin k:=k+1; a[k]:=j end
                        else k:=k+1;
if k=2 then writeln (a[1], ' ',a[2]);
                end;
end.
```

Напишите программу, которая ищет среди целых чисел, принадлежащих числовому отрезку [228224; 531135], числа, среди делителей которых есть хотя бы 4 различных **куба натуральных нечетных чисел**. Для каждого найденного числа запишите количество таких делителей, наибольший из них и само число. В качестве делителей не рассматривать число 1. Так, например, для числа 8 учитываются делители 2, 4, 8.

*Например, для числа 54 имеем следующие делители 2, 3, 6, 9, 18, 27, 54. Есть только один делитель – куб натурального числа $27=3*3*3$*

```
var i,j,k,m,d:integer;
begin
  for i:=228224 to 531135 do begin
    d:=3; k:=0; m:=0;
    while d*d*d<=i do begin
      if i mod (d*d*d)=0 then begin k:=k+1; m:=d*d*d; end;
      d:=d+2;
    end;
    if k>3 then writeln(k, ' ', m, ' ', i);
  end;
end.
```

Программы для написания программ

Компилятор онлайн:

https://rextester.com/l/pascal_online_compiler

Скачать PascalABC.NET:

<http://pascalabc.net/ssyilki-dlya-skachivaniya>

Д/З

<http://kpolyakov.spb.ru/school/ege.htm>

16 ноября Д/З (ЕГЭ22-№118,154,156)!

Сдающим ЕГЭ по информатике к этому+78
и 95

- [ЕГЭ22](#)

На следующий урок – на 23 ноября:

- [ЕГЭ25](#) № 33,57